

Resolint User's Guide

Introduction

Resolint is an OSATE-based linter tool for AADL models. Resolint provides a language for specifying rules that correspond to modeling guidelines, as well as a checker for evaluating whether a model complies with the rules. Results of the Resolint analysis are displayed to the user. Rule violations will indicate severity, and are linked to the model element that is out of compliance with the rule.

How to Get Resolint

Resolint is currently included with the Resolute plugin for OSATE, which can be found here:

<https://github.com/loonwerks/formal-methods-workbench/tree/master/tools/resolute>

Relationship with Resolute

Resolint statements are contained in the Resolute annexes of AADL models. This is because Resolint rules are specified using the same grammar as Resolute claims. In addition, the Resolute evaluation engine is used to determine whether the AADL model is in compliance with the Resolint rules.

Otherwise, Resolint and Resolute are two different tools with two different use cases. Future versions of Resolint may have greater independence from Resolute.

Formalizing Rules in Resolint

Rules derived from sources such as development standards, checklists, and modeling guidelines can be encoded in Resolint and embedded in a Resolute Annex in an AADL model. Rules in Resolint are represented using the same syntax as Resolute claims. For more information on specifying claims in Resolute, see the Resolute User's Guide here:

https://github.com/loonwerks/formal-methods-workbench/blob/master/documentation/resolute/user_guide.pdf

For example, the following rule

Threads should have the Dispatch_Protocol property specified

would be specified in Resolint as

```
annex resolute {**
    dispatch_protocol_specified() <=
        ** "Threads should have the Dispatch_Protocol property specified" **
        forall (t : thread) .
            lint_check(t, has_property(t, Dispatch_Protocol))
**};
```

Similarly, the rule

Threads can only specify a Dispatch_Property of Periodic or Sporadic

Would be specified as

```
annex resolute {**
    valid_dispatch_protocol() <=
        ** "Threads can only specify a dispatch_protocol property of periodic or sporadic" **
        forall (t : thread) . lint_check(t, has_property(t, Dispatch_Protocol) =>
            (property(t, Dispatch_Protocol) = "Sporadic" or property(t, Dispatch_Protocol) = "Periodic"))
**};
```

Note that each of these rules contains a call to *lint_check()*. *lint_check()* is a provided function that enables Resolint to capture the specific model element that violates the rule. The definition of *lint_check()* is specified in Resolint.aadl, which is included with the tool.

```
lint_check(a : aadl, b : bool) <=
    ** a **
    b
```

The function takes an AADL element and a Boolean value. The Boolean value represents the result of the rule check. If it is false, Resolint keeps track of the AADL element that violated the rule in order to provide the user with a direct reference in the results pane.

Two other check functions are provided: *lint_check_set()* and *lint_check_list()*. These are used when multiple elements are referenced in a rule. For example, the rule

```
annex resolute {**
    one_process() <=
        ** "Only one processor-bound process can contain thread or thread-group subcomponents" **
        let proc : {process} = {p for (p : process) |
            (exists(pr : processor) . is_bound_to(p, pr)) and forall(s : subcomponents(p)) . (is_thread(s) or is_thread_group(s))};
        lint_check_set(proc, size(proc) = 1)
**};
```

will be violated if multiple process components exist that contain threads or thread groups. If this is the case, the user should be presented with the set of all such processes. *lint_check_set()* evaluates the size of the set of processes containing threads, and if not equal to 1, will flag all processes in the set.

The *lint_check()* functions are not necessary for Resolint to check rules and display results. However, they are currently necessary to link results with the AADL elements that are violating the rules. This enables users to easily navigate to the AADL component that is violating a given rule by clicking on the rule violation in the Problems pane. Future versions of Resolint may eliminate the need for the *lint_check()* functions.

Creating Rulesets

Rules can be grouped into Rulesets. These are useful for organizing rules corresponding to different guidelines, such as organizational process, customer requirements, certification guidelines, and tool constraints. Rulesets also provide the ability to specify the severity of the rule violation; that is, if the rule is found to be violated, what type of message should the user receive. Three levels of severity are supported. From least to most severe, they are *info*, *warning*, and *error*.

The syntax of a ruleset is

`<Ruleset> ::= 'ruleset' <name> '{' (<Lint_Statement>)* '}'`

`<Lint_Statement> ::= (('info' | 'warning' | 'error') <Claim_Function_Reference>)*`

where `<Claim_Function_Reference>` is the name of a Resolute claim representing the rule.

Lint Statements are interpreted such that if the Claim Function is false, the user will receive a message marker of the severity indicated by the info, warning, or error keyword.

An example Ruleset is below.

```
annex resolute {**
  ruleset CASE_Tools {
    info (print("Linting CASE_Tools ruleset"))

    -- Threads should have the Dispatch_Protocol property specified
    warning (dispatch_protocol_specified())
    -- Threads can only specify a dispatch_protocol property of periodic or sporadic
    error (valid_dispatch_protocol())

    -- Subcomponent types should be specified
    warning (subcomponent_type_specified())

    -- Array dimensions must be specified
    error (array_dimension())
    -- Arrays can only have one dimension
    error (one_dimensional_arrays())

    -- The array base type should be specified
    warning (array_base_type())
    -- Connections between thread components must be uni-directional
    error (unidirectional_connections())
    -- A processor's subcomponents may be ignored
    warning (no_processor_subcomponents())
    -- AADL modes are not currently supported by CASE tools
    warning (no_modes_in_model())
  }
**};
```

Checking Rules and Rulesets

In order to check that an AADL model complies with a set of rules, Resolint needs to know which rules or rulesets to check. This is specified using the `check()` function call in an AADL component implementation. For example, the `check (CASE_Tools)` call in the UAS.Impl component pictured below will evaluate the UAS.Impl system instance against the CASE_Tools ruleset.

```
system UAS
end UAS;

system implementation UAS.Impl
  subcomponents
    GND: system GS::GroundStation.Impl;
    UAV: system UAV::UAV.Impl;
    RFA: bus UAS_Buses::RF_Bus.Impl;
  connections
    c1: port GND.radio_send -> UAV.radio_rcv;
    c2: port UAV.radio_send -> GND.radio_rcv;
    bac1: bus access RFA <-> GND.RFA;
    bac2: bus access RFA <-> UAV.RFA;
  properties
    Actual_Connection_Binding => (reference (RFA)) applies to c1, c2;

  annex resolute {**
    -- Make sure this model complies with the CASE toolchain modeling guidelines
    check (CASE_Tools)
  **};
end UAS.Impl;
```

Similarly, individual rules that do not belong to a Ruleset can also be checked, as in the following example:

```
system UAS
end UAS;


system implementation UAS.Impl
subcomponents
  GND: system GS::GroundStation.Impl;
  UAV: system UAV::UAV.Impl;
  RFA: bus UAS_Buses::RF_Bus.Impl;
connections
  c1: port GND.radio_send -> UAV.radio_rcv;
  c2: port UAV.radio_send -> GND.radio_rcv;
  bac1: bus access RFA <-> GND.RFA;
  bac2: bus access RFA <-> UAV.RFA;
properties
  Actual_Connection_Binding => (reference (RFA)) applies to c1, c2;

annex resolute {**
  -- Only one processor-bound process can contain thread or thread-group subcomponents
  check ( error( one_process() ) )
**};

end UAS.Impl;
```


Running Resolint

To run Resolint, select an AADL component implementation containing a *check()* statement in a Resolute annex. From the main menu, select Analyses → Resolint.



Resolint Output

When the Resolint analysis is complete a message box will inform the user whether any rule violations were discovered, and if so, how many of each severity type:



In addition, the list of rule violations will appear in the standard OSATE 'Problems' pane:

The screenshot displays the OSATE IDE interface. The top pane shows the AADL model 'UAS.aadl' with the following code:

```

1 package UAS
2 public
3   with GS;
4   with UAV;
5   with UAS_Buses;
6
7   -- UAS
8   system UAS
9   end UAS;
10
11 system implementation UAS.Impl1
12   subcomponents
13     GND: system GS::GroundStation.Impl1;
14     UAV: system UAV::UAV.Impl1;
15     RFA: bus UAS_Buses::RF_Bus.Impl1;
16   connections
17     c1: port GND.radio_send -> UAV.radio_rcv;
18     c2: port UAV.radio_send -> GND.radio_rcv;
19     bac1: bus access RFA <-> GND.RFA;
20     bac2: bus access RFA <-> UAV.RFA;
21   properties
22     Actual_Connection_Binding => (reference (RFA)) applies to c1, c2;
23
24 annex resolute {**
25   -- Make sure this model complies with the CASE toolchain modeling guidelines
26   check ( CASE_Tools )
27   **};
28
29 end UAS.Impl1;
30
31 end UAS;

```

The bottom pane shows the 'Problems' view with 3 errors, 2 warnings, and 0 others. The table below represents the data shown in this pane:

Description	Resource	Path	Location	Type
<ul style="list-style-type: none"> Components bound to a virtual processor may only communicate with components bound to other processes 	Modeling_Gui...	/CASE_Simple_...	line 37	Resolint
<ul style="list-style-type: none"> Only one processor-bound process can contain thread or thread-group subcomponents 	GS.aadl	/CASE_Simple_...	line 78	Resolint
<ul style="list-style-type: none"> Only one processor-bound process can contain thread or thread-group subcomponents 	MC.aadl	/CASE_Simple_...	line 106	Resolint
<ul style="list-style-type: none"> Data types should be specified 	Modeling_Gui...	/CASE_Simple_...	line 20	Resolint
<ul style="list-style-type: none"> Threads should have the Dispatch_Protocol property specified 	SW.aadl	/CASE_Simple_...	line 228	Resolint

Double-clicking on an individual problem will open the file containing the AADL element violating the rule and highlight it, as well as place a marker with the corresponding severity in the margin:

UAS.aadl
Modeling_Guidelines.aadl
GS.aadl

```

64         features
65             radio_recv: in event data port;
66             radio_send: out event data port;
67             RFA: requires bus access UAS_Buses::RF_Bus.Impl;
68         properties
69             CASE_Properties::BOUNDARY => (PHYSICAL);
70     end GroundStation;
71
72     system implementation GroundStation.Impl
73     subcomponents
74         RADIO_HW: device Radio.Impl;
75         PROC_HW: processor GS_Proc.Impl;
76         MEM_HW: memory GS_Mem.Impl;
77         BUS_HW: bus GS_Bus.Impl;
78         PROC_SW: process GS_SW.Impl;
79     connections
80         bac1: bus access RADIO_HW.GSA <-> BUS_HW;
81         bac3: bus access PROC_HW.GSA <-> BUS_HW;
82         bac4: bus access MEM_HW.GSA <-> BUS_HW;
83         bac6: bus access RADIO_HW.RFA <-> RFA;
84         c1: port PROC_SW.send_data -> RADIO_HW.send_data_in;
85         c2: port RADIO_HW.send_data_out -> radio_send;
86         c3: port radio_recv -> RADIO_HW.recv_data_in;
87         c4: port RADIO_HW.recv_data_out -> PROC_SW.recv_data;
88     properties
89         Actual_Processor_Binding => (reference (PROC_HW)) applies to PROC_SW;
90         Actual_Memory_Binding => (reference (MEM_HW)) applies to PROC_SW;
91         Actual_Connection_Binding => (reference (BUS_HW)) applies to c1, c4;
92     end GroundStation.Impl;
93
94 end GS;

```

Problems
Properties
Console

3 errors, 2 warnings, 0 others

Description	Resource	Path	Location	Type
Errors (3 items)				
Components bound to a virtual processor may only communicate with components bound to other proces	Modeling_Gui...	/CASE_Simple_...	line 37	Resolint
Only one processor-bound process can contain thread or thread-group subcomponents	GS.aadl	/CASE_Simple_...	line 78	Resolint
Only one processor-bound process can contain thread or thread-group subcomponents	MC.aadl	/CASE_Simple_...	line 106	Resolint
Warnings (2 items)				
Data types should be specified	Modeling_Gui...	/CASE_Simple_...	line 20	Resolint
Threads should have the Dispatch_Protocol property specified	SW.aadl	/CASE_Simple_...	line 228	Resolint